บทที่ 2

รากของสมการ

สมการ $ax^2 + bx + c = 0$ จะมีรากของสมการคือ $x = \dfrac{-b \pm \sqrt{b^2 - 4ac}}{2a}$ ซึ่งสามารถที่จะหาค่าได้ง่าย ๆ แต่ยังมีสมการอื่นอีกมากมายซึ่งไม่สามารถหารากได้ในลักษณะเช่นนี้ ก่อนที่จะมีการประดิษฐ์คอมพิวเตอร์ขึ้น มีวิธีการหลายวิธีในการหารากของสมการ บางกรณีสามารถคำนวณรากได้โดยตรง เช่น ตัวอย่างข้างบน แต่บางกรณีไม่สามารถหารากได้ เช่น ฟังก์ชัน $f(x) = e^{-x} - x$ ซึ่งจะต้องหารากโดยการประมาณค่า วิธีการหนึ่งที่ใช้กันคือวิธีกราฟ โดยการเขียนกราฟฟังก์ชัน f(x) กับค่า x จุดที่กราฟตัดแกน x จะเป็นค่ารากของสมการ ที่จุดนี้ f(x) = 0 ซึ่งเราจะได้ศึกษากันอย่างละเอียดต่อไป

2.1 วิธีกราฟ (Graphical method)

วิธีการอย่างง่ายในการหารากของสมการ f(x) = 0 ก็คือวิธีกราฟ โดยการเขียนกราฟค่าของฟังก์ชันทางแกน y และ ตัวแปร x ทางแกน x จุดที่กราฟตัดแกน x จะเป็นค่ารากของสมการ การหาค่ารากของสมการโดยวิธีกราฟเช่นนี้ จะได้ค่ารากโดยประมาณเท่านั้น แต่ก็เป็นวิธีการที่มีประโยชน์มากในการตรวจสอบว่ารากที่หาด้วยวิธีการอื่น มีความถูกต้องใกล้เคียงกับค่าที่แท้จริงหรือไม่

ตัวอย่างที่ 2.1 จงอาศัยวิธีกราฟหาค่าสัมประสิทธิ์ c เมื่อนักกระโดดร่มมวล m=68.1 กิโลกรัม กระโดดลงมาจากลูกบอลลูนที่ลอยอยู่นิ่ง โดยมีความเร็ว 40 เมตร/วินาที เมื่อเวลาผ่านไป 10 วินาที

วิธีทำ จากกฎข้อที่สองของนิวตัน ความเร็วของนักกระโดดร่มเมื่อเวลาผ่านไป t วินาที คือ

$$v(t) = \frac{gm}{c}(1 - e^{-(c/m)t})$$

กำหนดให้

$$f(c) = \frac{gm}{c}(1 - e^{-(c/m)t}) - v$$

จากโจทย์ t = 10, g = 9.8, v = 40 และ m = 68.1

$$f(c) = \frac{9.8(68.1)}{c}\left(1 - e^{-(c/68.1)10}\right) - 40$$

$$f(c) = \frac{667.38}{c}\left(1 - e^{-0.146843c}\right) - 40$$

แทนค่า c ต่าง ๆ กัน เพื่อที่จะนำไปเขียนกราฟ โดยใช้ค่า c ที่ทำให้ f(c) เป็นทั้งค่าบวกและค่าลบ

c	$f(c)$
4	34.115
8	17.653
12	6.067
16	-2.269
20	-8.401

นำค่าในตารางไปเขียนกราฟ จะได้กราฟดังรูป 2.1



รูปที่ 2.1 กราฟระหว่าง f(c) และ c เพื่อหาค่ารากของสมการ

จากกราฟจะเห็นได้ว่า c ตัดแกน x ที่จุดระหว่าง x = 12 และ x = 16 จากการประมาณด้วยสายตาจะได้ค่า c ประมาณ 15 เพื่อการตรวจสอบ แทนค่า c = 15 ลงในสมการ จะได้

$$f(15) = \frac{667.38}{15}\left(1-e^{-0.146843(15)}\right) - 40$$
$$= -0.4248319$$

ซึ่งมีค่าใกล้กับศูนย์ แต่ไม่ตรงทีเดียว เนื่องจากว่าเป็นการประมาณด้วยสายตา

2.2 วิธีแบ่งส่วน (Bisection method)



รูปที่ 2.2 วิธีแบ่งส่วน

วิธีแบ่งส่วนเป็นวิธีการง่ายที่สุดในการหารากของสมการ สามารถกระทำได้โดยใช้เครื่องคิดเลขหรืออาจเขียนโปรแกรมคอมพิวเตอร์มาช่วยในการคำนวณ ขั้นตอนในการคำนวณมีดังต่อไปนี้

ขั้นที่ 1 เลือกค่าต่ำ x_l และค่าสูง x_u โดยมีค่ารากอยู่ระหว่างค่าทั้งสอง นั่นคือ

$$f(x_l)f(x_u) < 0$$

ขั้นที่ 2 ประมาณค่าราก x_r โดย

$$x_r = \frac{x_l + x_u}{2}$$

ขั้นที่ 3 หาว่ารากอยู่ระหว่างช่วงใด

ก) ถ้า $f(x_l)f(x_u) < 0$ รากจะอยู่ในช่วงล่างให้ ตั้งค่า $x_u = x_r$ แล้วกลับไปขั้นที่ 2
ข) ถ้า $f(x_l)f(x_u) > 0$ รากจะอยู่ในช่วงบนให้ ตั้งค่า $x_l = x_r$ แล้วกลับไปขั้นที่ 2
ค) ถ้า $f(x_l)f(x_u) = 0$ รากจะเท่ากับ x_r ให้หยุดโปรแกรม

ทำตามขั้นตอนดังกล่าวซ้ำอีกหลาย ๆ ครั้ง จนกระทั่ง ค่ารากมีค่าใกล้เคียงกับค่ารากที่แท้จริง

ตัวอย่างที่ 2.2 ใช้วิธีแบ่งส่วน หาค่า c ในตัวอย่างที่ผ่านมา

วิธีทำ ขั้นแรกเดาค่า c สองค่า ซึ่งทำให้ $f(c)$ มีเครื่องหมายตรงกันข้ามในที่นี้ได้ 12 และ 16
ดังนั้น

$$x_r = \frac{12+16}{2} = 14$$

เป็นค่าโดยประมาณ มีความผิดพลาด ε_t = 5.3 % (เทียบกับค่าที่แท้จริงของราก คือ 14.7802)

ขั้นต่อไปคำนวณผลคูณ

$$f(12)f(14) = 6.067(1.569) = 9.517$$

ซึ่งมีค่ามากกว่าศูนย์ ไม่มีการเปลี่ยนเครื่องหมายระหว่างจุดต่ำและจุดกึ่งกลาง ดังนั้นราก จะอยู่ระหว่าง 14 และ 16

$$x_r = \frac{14+16}{2} = 15$$

ซึ่งจะมีความคลาดเคลื่อน ε_t = 1.5 %

เราสามารถทำซ้ำอีก คือ

$$f(14)f(15) = 1.569(-0.425) = -0.666$$

ดังนั้น รากจะอยู่ระหว่าง 14 และ 15

จะได้

$$x_r = \frac{14+15}{2} = 14.5$$

ซึ่งจะมีค่าความคลาดเคลื่อน ε_t = 1.9 %

นักศึกษาสามารถทำซ้ำหลาย ๆ ครั้ง จนกว่าจะได้ความถูกต้องตามที่ต้องการ

ตัวอย่างที่ 2.3 จงเขียนโปรแกรมภาษาฟอร์แทรนเพื่อหารากของสมการ $-0.4x^2+2.2x+4.7$ โดยวิธีแบ่งส่วน ใช้ $x_l = 5$ และ $x_u = 10$

วิธีทำ

```
*-- program bisec
*--
*-- The interval a<x<b is known to contain a root of f(x).the
*-- estimate of the root is successively improved by finding in
*-- which half of the interval the root lies and then replacing
*-- the original interval by that half.
*-----------------------------------------------------------
```

```
*Variables
*--
          REAL X1,X2,X3,F1,F2,F3,A,B,EPS,D,DO
          INTEGER I,IMAX
*--
*--       X1,X3,X2 -- The left,right,and midpoint ot the
*--                        currert interval
*--       F1,F3,F2 -- The function evaluated at these  points
*--       A,B,DO    -- The left and right ends of the original
*--                        interval and its width (b-a)
*--         EPS     -- convergence criterion  based on the size of
*--                         the current interval
*--          D      -- The widih of the current interval (x3-x1)
*--         IMAX    -- Maximum number of iterrations
*--          I      -- Current iteration counter
*--------------------------------------------------------------
*Statement function for the function f(x)
*--
       F(X) = -0.4*X*X+2.2*X+4.7
*--
*--         (or any other function)
*--------------------------------------------------------------
* Initialization
*--
       PI = ACOS(-1.)
       WRITE(*,*)'Enter limits of original search interval, a,b'
       READ (*,*)A,B
       WRITE(*,*)'Enter convergence criterion (EPS) and',
     +           'max.no.of iterations (IMAX)'
```

```
      READ (*,*)EPS,IMAX
*--
      WRITE(*,*)'The original search interval is from',A,'to',B
      WRITE(*,*)'The convergence criterion is (interval)<',EPS
      WRITE(*,*)'The maximum No. of iterations allowed is ',IMAX
      WRITE(*,*)
*--
      X1 = A
      X3 = B
      F1 = F(X1)
      F3 = F(X3)
      DO = (B-A)
*--
*--     First verify that there is  indeed a root in the interval.
*--
      IF(F1*F3.GT.0.0)THEN
        WRITE(*,*)'No root in original interval. It is possible'
        WRITE(*,*)'that the function is incorrectly coded.'
        ELSE
*     Begin Iterations loop
*--
        DO 3I = 1,IMAX
*--
*--     Find which half of the interval containsthe root
*--
        X2 = .5*(X1+X3)
        F2 = F(X2)
        IF(F1*F2.LE.0.0)THEN
*--
```

```
*--         Root is in left half, so
*--
             D = (X2-X1)/2.
             F3 = F2
             X3 = X2
          ELSE
*--
*--         Root is in right half,so
*--
            D = (X3-X2)/2.
            F1 = F2
            X1 = X2
          ENDIF
*--
*--       Test for convergence
*--
          IF(D .LT.EPS)THEN
*--
*--       Success path
*--
          WRITE(*,*)'A root at X = ',X2,'was foud'
          WRITE(*,*)'in',I,'iterations'
          WRITE(*,*)'The value of the function is ',F2
          STOP 'success'
        ENDIF
3       CONTINUE
*----------------------------------------------------
*     Excessive Iterations Path
*--
```

```
 WRITE(*,*)'After ',IMAX,' iterations, no root was fouad'
 WRITE(*,*)' Within the convergence criterion'
 STOP 'failure'
ENDIF
END
```

เอาท์พุทจากโปรแกรม คือ

```
 Enter limits of original search interval, a,b
5
10
 Enter convergence criterion (EPS) andmax.no.of iterations (IMAX)
.01
20
 The original search interval is from  5.to  10.
 The convergence criterion is (interval)<  0.00999999978
 The maximum No. of iterations allowed is  20

 A root at X =   7.12890625was foud
 in 8iterations
 The value of the function is   0.055071868
STOP success statement executed
```

ดังนั้น รากของสมการ คือ 7.12890625

ตัวอย่างที่ 2.4 การประมาณความคลาดเคลื่อนของวิธีแบ่งส่วนให้นักศึกษาทำซ้ำตามตัวอย่าง 2.2 จนกระทั่งความคลาดเคลื่อนน้อยกว่า ε_s = 0.5 %

วิธีทำ จากตัวอย่าง 2.2 ราก คือ 14 และ 15

ความคลาดเคลื่อนจากการประมาณ

$$|\varepsilon_a| = \left|\frac{15-14}{15}\right| 100\% = 6.667\%$$

ความคลาดเคลื่อนจริง ε_t = 1.5 % ดังนั้น $\varepsilon_a > \varepsilon_t$

iteration	x_l	x_u	x_r	$\|\varepsilon_a\|$	$\|\varepsilon_t\|$
1	12	16	14		5.279
2	14	16	15	6.667	1.487
3	14	15	14.5	3.448	1.896
4	14.5	15	14.75	1.695	0.204
5	14.75	15	14.875	0.840	0.641
6	14.75	14.875	14.8125	0.422	0.219

จากตารางจะเห็นว่าหลังจาก iteration ครั้งที่ 6 $\varepsilon_a < 0.5\%$

ตัวอย่างที่ 2.5 จงคำนวณรากของสมการ f(x) = $\ln x^2 - 0.7$ โดยวิธีแบ่งส่วนใช้ $x_l = 0.5$ และ $x_u = 2$

วิธีทำ ใช้โปรแกรมตามตัวอย่าง 2.3 โดยเปลี่ยน f(x) = $\ln x^2 - 0.7$

เอาท์พุทจากโปรแกรม คือ

```
 Enter limits of original search interval, a,b
0.5
2
 Enter convergence criterion (EPS) andmax.no.of iterations (IMAX)
.05
10
 The original search interval is from  0.5to  2.
 The convergence criterion is (interval)<  0.0500000007
 The maximum No. of iterations allowed is  10

 A root at X =   1.34375was found
 in 4 iterations
 The value of the function is  -0.109071553
STOP success statement executed
```

รากของสมการคือ 1.34375

2.3 วิธีฟอลส์โพซิชัน (False-Position Method)

วิธีฟอลส์โพซิชัน หรือ เร็กกูลาฟอลไซ(regula falsi) หรือ การอินเทอโพเลทเชิงเส้น (linear interpolation) หาจากความชันของกราฟ

$$\frac{f(x_i)}{x_r - x_l} = \frac{f(x_u)}{x_r - x_u}$$

$$f(x_l)(x_r - x_u) = f(x_u)(x_r - x_l)$$

$$x_r\left[f(x_l) - f(x_u)\right] = x_u f(x_l) - x_l f(x_u)$$

หารด้วย $f(x_l) - f(x_u)$

$$x_r = \frac{x_u f(x_l) - x_l f(x_u)}{f(x_l) - f(x_u)}$$

$$x_r = \frac{x_u f(x_l)}{f(x_l) - f(x_u)} - \frac{x_l f(x_u)}{f(x_l) - f(x_u)}$$

$$x_r = x_u + \frac{x_u f(x_l)}{f(x_l) - f(x_u)} - x_u - \frac{x_l f(x_u)}{f(x_l) - f(x_u)}$$

$$x_r = x_u + \frac{x_u f(x_u)}{f(x_l) - f(x_u)} - \frac{x_l f(x_u)}{f(x_l) - f(x_u)}$$

หรือ

$$x_r = x_u - \frac{f(x_u)(x_l - x_u)}{f(x_l) - f(x_u)}$$

จะได้

$$x_r = x_u - \frac{f(x_u)(x_l - x_u)}{f(x_l) - f(x_u)}$$

ใช้เป็นสูตรในการคำนวณหารากของสมการ



รูปที่ 2.3 วิธีฟอลส์โพซิชัน

ตัวอย่างที่ 2.6 จงใช้วิธีฟอลส์โพซิชัน หารากของสมการในตัวอย่าง 2.1

วิธีทำ เดา $x_l = 12$ และ $x_u = 16$

<u>first iteration</u>

$x_l = 12$ $f(x_l)$ = 6.0069

$x_u = 16$ $f(x_u)$ = -2.2688

ดังนั้น $$x_r = 16 - \frac{-2.2688(12-16)}{6.0669-(-2.2688)} = 14.9113$$

ซึ่งมี True error ε_t = 0.89 %

second iteration

$f(x_l)\,f(x_r) = -1.5426$

ดังนั้นรากอยู่ในช่วงแรก

$x_l = 12$ $f(x_l)$ = 6.0069

$x_u = 14.9113$ $f(x_u)$ = -0.2543

$$x_r = 14.9113 - \frac{-2.2688(12-14.9113)}{6.0669-(-0.2543)} = 14.7942$$

ซึ่งมีความคลาดเคลื่อนจริง = 0.09 และความคลาดเคลื่อนประมาณ = 0.79 % ถ้าต้องการความถูกต้องมากกว่านี้ ให้ทำการ iterate ต่อไป

จะเห็นได้ว่า วิธี false – position จะได้ค่ารากของสมการเร็วกว่าวิธีแบ่งส่วนมาก กราฟรูปที่ 2 เปรียบเทียบ % ความคลาดเคลื่อนกับจำนวนครั้งของการ iterate โดยวิธี false – postion กับวิธีแบ่งส่วน



รูปที่ 2.4 เปรียบเทียบความคลาดเคลื่อน

ข้อเสียของวิธีฟอลส์โพซิชัน

ถึงแม้ว่าวิธีฟอลส์โพซิชัน จะให้คำตอบได้เร็วกว่าวิธีแบ่งส่วน แต่ในบางกรณี วิธีฟอลส์โพซิชันอาจจะไม่ให้คำตอบเลย ดังตัวอย่างต่อไปนี้

ตัวอย่างที่ 2.7 ใช้วิธีแบ่งส่วนและวิธีฟอลส์โพซิชันหาค่ารากของสมการ

$$f(x) = x^{10} - 1$$

ระหว่าง $x = 0$ และ 1.3

วิธีทำ a) วิธีแบ่งส่วน

iteration	x_l	x_u	x_r	$\lvert\varepsilon_t\rvert$	$\lvert\varepsilon_a\rvert$
1	0	1.3	0.65	35	100.0
2	0.65	1.3	0.975	2.5	33.3
3	0.975	1.3	1.1375	13.8	14.3
4	0.975	1.1375	1.05625	5.6	7.7
5	0.975	1.05625	1.015625	1.6	4.0

b) วิธีฟอลส์โพซิชัน

iteration	x_l	x_u	x_r	$\lvert\varepsilon_t\rvert$	$\lvert\varepsilon_a\rvert$
1	0	1.3	0.09430	90.6	
2	0.09430	1.3	0.18176	81.8	48.1
3	0.18176	1.3	0.26287	73.7	30.9
4	0.26287	1.3	0.33811	66.2	22.3
5	0.33811	1.3	0.40788	59.2	17.1

จะเห็นได้ว่าสำหรับฟังก์ชันนี้ วิธีฟอลส์โพซิชันเข้าถึงคำตอบได้ช้ากว่าวิธีแบ่งส่วนมาก ทำซ้ำถึง 5 ครั้ง ε_t ยังมีค่ามากถึง 59.2 % ก็คือยังไม่ได้คำตอบที่ถูกต้อง

```
      PROGRAM MOD_FALSE_POSITION
      WRITE(*,*)'ENTER XL,XR,ES'
      READ (*,*)XL,XR,ES
      FXL= FUNC(XL)
      FXR= FUNC(XR)
      TEST = FXL*FXR
      IF(TEST.GE.0.) THEN
        WRITE(6,10)
        STOP
      ENDIF
10     FORMAT(/, ' ROOT IS OUT OF RANGE')
      XD = XL
      WRITE(6,20)
20     FORMAT(/, 3X, 'ITERATION NO. ', 9X, 'X', /)
      DO 100 ITER=1,3000
      FXL = FUNC(XL)
```

```
      FXR = FUNC(XR)
      XFP  = (XL*FXR - XR*FXL)/(FXR - FXL)
      FX1 = FUNC(XFP)
      TEST  = FX1*FXR
        IF(TEST.LT.0.)  THEN
           XL = XFP
        ELSE
           XR = XFP
        ENDIF
      WRITE(6,30) ITER, XFP
30     FORMAT(1X, I8, 8X, E14.6)
      TL = ABS((XFP-XD)*100./XFP)
      IF(TL.LT.ES) GO TO 50
      XD = XFP
100    CONTINUE
      WRITE(6,40)
40    FORMAT(/, ' ROOT CAN NOT FIND IN THIS CONDITION')
      GO TO 200
50    WRITE(6,60) XFP
60    FORMAT(/, 3X, 'THE ROOT IS ', E14.6)
200    CONTINUE
      STOP
      END
*---------------------------------------------------
      FUNCTION FUNC(X)
      FUNC = (X**10)-1
      RETURN
      END
```

เอาท์พุทจากโปรแกรม คือ

```
 ENTER XL,XR,ES
0
1.3
.00001

   ITERATION NO.            X

        1            0.942996E-01
        2            0.181759E+00
        3            0.262874E+00
        4            0.338105E+00
        5            0.407878E+00
        6            0.472583E+00
        7            0.532572E+00
        8            0.588145E+00
        9            0.639544E+00
       10            0.686943E+00
```

```
       52            0.999990E+00
       53            0.999993E+00
       54            0.999994E+00
       55            0.999996E+00
       56            0.999997E+00
       57            0.999997E+00
       58            0.999998E+00
       59            0.999999E+00
       60            0.999999E+00
       61            0.999999E+00
       62            0.999999E+00
       63            0.100000E+01
       64            0.100000E+01
       65            0.100000E+01

   THE ROOT IS    0.100000E+01
```

รากของสมการ คือ 1.0 โดยทำการ iterate ถึง 63 ครั้งจึงจะได้คำตอบถูกต้อง



รูปที่ 2.5 กราฟ $f(x) = x^{10} – 1$ แสดงการลู่ของวิธีฟอลส์โพซิชัน

ตัวอย่างที่ 2.8 จงเขียนโปรแกรมภาษาฟอร์แทรนเพื่อหารากของสมการ

$$f(x) = -26 + 82.3x - 88x^2 + 45.4x^3 - 9x^4 + 0.65x^5$$

โดยวิธีฟอลส์โพซิชัน ใช้ $x_l = 0.5$, $x_a = 1$

วิธีทำ โปรแกรมที่เขียนโดยวิธีฟอลส์โพซิชัน คือ

```
      PROGRAM FALSE_POSITION
      WRITE(*,*)'ENTER XL,XR,ES'
      READ (*,*)XL,XR,ES
      FXL= FUNC(XL)
      FXR= FUNC(XR)
      AA = FXL*FXR
      IF(AA.GE.0.) THEN
        WRITE(6,10)
        STOP
      ENDIF
10     FORMAT(/, ' ROOT IS NOT IN THE GIVEN RANGE')
      X10LD = XL
```

```
      DO 100 I=1,500
      FXL = FUNC(XL)
      FXR = FUNC(XR)
      X1  = (XL*FXR - XR*FXL)/(FXR - FXL)
      FX1 = FUNC(X1)
      AA  = FX1*FXR
        IF(AA.LT.0.)  THEN
          XL = X1
        ELSE
          XR = X1
        ENDIF
      TOL = ABS((X1-X10LD)*100./X1)
      IF(TOL.LT.ES) GO TO 40
      X10LD = X1
100     CONTINUE
      WRITE(6,20)
20      FORMAT(/, ' ROOT CAN NOT BE REACHED FOR',
    +          ' THE GIVEN CONDITION'       )
      GO TO 50
40      WRITE(6,30) X1
30      FORMAT(/, 3X, 'THE ROOT IS ', E14.6)
50      CONTINUE
      STOP
      END
      FUNCTION FUNC(X)
      FUNC = -26+82.3*X-88*X*X+45.4*X**3-9*X**4+0.65*X**5
      RETURN
      END
```

เอาท์พุทจากโปรแกรม คือ

```
 ENTER XL,XR,ES
0.5
1
.05

   THE ROOT IS    0.579331E+00
```

ดังนั้น รากของสมการ คือ 0.579331

ตัวอย่างที่ 2.9 ความเร็วของนักกระโดดร่มเป็นไปตามสมการ

$$v(t) \quad = \quad \frac{gm}{c}(1 - e^{-(c/m)t})$$

เมื่อ g = 9.8, c = 14 กก./วินาที จงหามวล m ของนักกระโดดร่ม ซึ่งมีความเร็ว v = 35 เมตร/วินาที เมื่อเวลาผ่านไป 7 วินาที ใช้วิธีฟอลส์โพซิชันที่มีความถูกต้องถึง 0.01%

วิธีทำ ใช้โปรแกรมตามตัวอย่าง 2.8 โดยเปลี่ยน f(x) เป็น v(t)

เอาท์พุทจากโปแกรม คือ

```
 ENTER XL,XR,ES
50
80
.01

   ITERATION NO.          X

        1           65.6020431519
        2           63.8795661926
        3           63.6767005920
        4           63.6528625488
        5           63.6500625610

   THE ROOT IS    63.6500625610
```

ดังนั้น มวลของนักกระโดดร่ม = 63.65 กิโลกรัม

ตัวอย่างที่ 2.10 จงหารากของสมการ $x^{3.3} = 79$ โดยวิธีฟอลส์โพซิชัน ใช้ $x_l = 3$, $x_u = 4$ และ $\varepsilon_s = 0.0001$

วิธีทำ ใช้โปรแกรมตามตัวอย่าง 2.8 โดยเปลี่ยน f(x)

เอาท์พุทจากโปรแกรม คือ

```
 ENTER XL,XR,ES
3
4
.0001
   ITERATION NO.           X

        1            3.6972043514
        2            3.7543210983
        3            3.7583978176
        4            3.7586855888
        5            3.7587058544
        6            3.7587072849

   THE ROOT IS      3.7587072849
```

ดังนั้น รากของสมการ คือ 3.7587072849

แบบฝึกหัดบทที่ 2

1. จงหารากของสมการ $f(x) = -2 + 7x - 5x^2 + 6x^3$
    ก) โดยวิธีกราฟ
    ข) โดยวิธีแบ่งส่วน โดยใช้ $x_l = 0$ และ $x_u = 1$ และทำหาร iterate จน ε_a น้อยกว่า 10%
2. จงหารากของสมการ $f(x) = -26 + 82.3x - 88x^2 + 45.4x^3 - 9x^4 + 0.65x^5$
    ก) โดยวิธีกราฟ
    ข) โดยวิธีแบ่งส่วน โดยใช้ $x_l = 0.5$ และ $x_u = 1.0$
    ค) โดยวิธีฟอลส์โพซิชัน
3. จงหารากของสมการ $f(x) = -11 - 22x + 17x^2 - 2.5x^3$
    ก) โดยวิธีกราฟ
    ข) โดยวิธีฟอลส์โพซิชัน